// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/proxy/Clones.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (proxy/Clones.sol)

pragma solidity ^0.8.0;

/**
 * @dev https://eips.ethereum.org/EIPS/eip-1167[EIP 1167] is a standard for
 * deploying minimal proxy contracts, also known as "clones".
 *
 * > To simply and cheaply clone contract functionality in an immutable way, this standard specifies
 * > a minimal bytecode implementation that delegates all calls to a known, fixed address.
 *
 * The library includes functions to deploy a proxy using either `create` (traditional deployment) or `create2`
 * (salted deterministic deployment). It also includes functions to predict the addresses of clones deployed using the
 * deterministic method.
 *
 * _Available since v3.4._
 */
library Clones {
    /**
     * @dev Deploys and returns the address of a clone that mimics the behaviour of `implementation`.
     *
     * This function uses the create opcode, which should never revert.
     */
    function clone(address implementation) internal returns (address instance) {
        /// @solidity memory-safe-assembly
        assembly {
            // Cleans the upper 96 bits of the `implementation` word, then packs the first 3 bytes
            // of the `implementation` address with the bytecode before the address.
            mstore(0x00, or(shr(0xe8, shl(0x60, implementation)), 0x3d602d80600a3d3981f3363d3d373d3d3d363d73000000))
            // Packs the remaining 17 bytes of `implementation` with the bytecode after the address.
            mstore(0x20, or(shl(0x78, implementation), 0x5af43d82803e903d91602b57fd5bf3))
            instance := create(0, 0x09, 0x37)
        }
        require(instance != address(0), "ERC1167: create failed");
    }

    /**
     * @dev Deploys and returns the address of a clone that mimics the behaviour of `implementation`.
     *
     * This function uses the create2 opcode and a `salt` to deterministically deploy
     * the clone. Using the same `implementation` and `salt` multiple time will revert, since
     * the clones cannot be deployed twice at the same address.
     */
    function cloneDeterministic(address implementation, bytes32 salt) internal returns (address instance) {
        /// @solidity memory-safe-assembly
        assembly {
            // Cleans the upper 96 bits of the `implementation` word, then packs the first 3 bytes
            // of the `implementation` address with the bytecode before the address.
            mstore(0x00, or(shr(0xe8, shl(0x60, implementation)), 0x3d602d80600a3d3981f3363d3d373d3d3d363d73000000))
            // Packs the remaining 17 bytes of `implementation` with the bytecode after the address.
            mstore(0x20, or(shl(0x78, implementation), 0x5af43d82803e903d91602b57fd5bf3))
            instance := create2(0, 0x09, 0x37, salt)
        }
        require(instance != address(0), "ERC1167: create2 failed");
    }

    /**
     * @dev Computes the address of a clone deployed using {Clones-cloneDeterministic}.
     */
    function predictDeterministicAddress(
        address implementation,
        bytes32 salt,
        address deployer
    ) internal pure returns (address predicted) {
        /// @solidity memory-safe-assembly
        assembly {
            let ptr := mload(0x40)
            mstore(add(ptr, 0x38), deployer)
            mstore(add(ptr, 0x24), 0x5af43d82803e903d91602b57fd5bf3ff)
            mstore(add(ptr, 0x14), implementation)
            mstore(ptr, 0x3d602d80600a3d3981f3363d3d373d3d3d363d73)
            mstore(add(ptr, 0x58), salt)
            mstore(add(ptr, 0x78), keccak256(add(ptr, 0x0c), 0x37))
            predicted := keccak256(add(ptr, 0x43), 0x55)
        }
    }

    /**
     * @dev Computes the address of a clone deployed using {Clones-cloneDeterministic}.
     */
    function predictDeterministicAddress(
        address implementation,
        bytes32 salt
    ) internal view returns (address predicted) {
        return predictDeterministicAddress(implementation, salt, address(this));
    }
}

// File: @openzeppelin/contracts/token/ERC20/ERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./extensions/IERC20Metadata.sol";
import "../../utils/Context.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.openzeppelin.com/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * The default value of {decimals} is 18. To change this, you should override
 * this function so it returns a different value.
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the default value returned by this function, unless
     * it's overridden.
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address to, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _transfer(owner, to, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * NOTE: If `amount` is the maximum `uint256`, the allowance is not updated on
     * `transferFrom`. This is semantically equivalent to an infinite approval.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * NOTE: Does not update the allowance if the current allowance
     * is the maximum `uint256`.
     *
     * Requirements:
     *
     * - `from` and `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     * - the caller must have allowance for ``from``'s tokens of at least
     * `amount`.
     */
    function transferFrom(address from, address to, uint256 amount) public virtual override returns (bool) {
        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        _transfer(from, to, amount);
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, allowance(owner, spender) + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        address owner = _msgSender();
        uint256 currentAllowance = allowance(owner, spender);
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(owner, spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `from` to `to`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     */
    function _transfer(address from, address to, uint256 amount) internal virtual {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(from, to, amount);

        uint256 fromBalance = _balances[from];
        require(fromBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[from] = fromBalance - amount;
            // Overflow not possible: the sum of all balances is capped by totalSupply, and the sum is preserved by
            // decrementing then incrementing.
            _balances[to] += amount;
        }

        emit Transfer(from, to, amount);

        _afterTokenTransfer(from, to, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        unchecked {
            // Overflow not possible: balance + amount is at most totalSupply + amount, which is checked above.
            _balances[account] += amount;
        }
        emit Transfer(address(0), account, amount);

        _afterTokenTransfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
            // Overflow not possible: amount <= accountBalance <= totalSupply.
            _totalSupply -= amount;
        }

        emit Transfer(account, address(0), amount);

        _afterTokenTransfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(address owner, address spender, uint256 amount) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Updates `owner` s allowance for `spender` based on spent `amount`.
     *
     * Does not update the allowance amount in case of infinite allowance.
     * Revert if not enough allowance is available.
     *
     * Might emit an {Approval} event.
     */
    function _spendAllowance(address owner, address spender, uint256 amount) internal virtual {
        uint256 currentAllowance = allowance(owner, spender);
        if (currentAllowance != type(uint256).max) {
            require(currentAllowance >= amount, "ERC20: insufficient allowance");
            unchecked {
                _approve(owner, spender, currentAllowance - amount);
            }
        }
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(address from, address to, uint256 amount) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * has been transferred to `to`.
     * - when `from` is zero, `amount` tokens have been minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(address from, address to, uint256 amount) internal virtual {}
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/interfaces/IMintableBurnableERC20.sol



// SPDX-License-Identifier: GPL-3.0

pragma solidity 0.8.17;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";

interface IMintableBurnableERC20 is IERC20 {
    function mint(address _to, uint256 _amount) external;
    function burn(uint256 _amount) external;
    function transferOwner(address _owner) external;
    function setMinter(address _minter, bool _status) external;
}

// File: contracts/interfaces/IVault.sol



// SPDX-License-Identifier: GPL-3.0

pragma solidity 0.8.17;

import "./IVaultFactory.sol";

interface IVault {
    function initialize(address) external;

    function factory() external view returns (IVaultFactory);
    function owner() external view returns (address);
    function deposited() external view returns (uint);
    function minted() external view returns (uint);
    function availableBalance() external view returns (uint);
    function pendingYield() external view returns (uint);
    function mintRatio() external view returns (uint);
    function protocolFee() external view returns (uint);
    function redemptionFee() external view returns (uint);

    function deposit(uint) external returns (uint);
    function withdraw(uint) external;
    function mint(uint) external;
    function burn(uint) external;
    function redeem(uint) external;
    function claim() external;
}

// File: contracts/interfaces/IVaultFactory.sol



// SPDX-License-Identifier: GPL-3.0

pragma solidity 0.8.17;

import "./IMintableBurnableERC20.sol";

interface IVaultFactory {
    function collateral() external view returns (IERC20);
    function token() external view returns (IMintableBurnableERC20);

    function protocolFeeTo() external view returns (address);
    function maxProtocolFee() external view returns (uint);
    function redemptionFeeTo() external view returns (address);
    function minRedemptionFee() external view returns (uint);
    function maxRedemptionAdjustment() external view returns (uint);
    function PRECISION() external view returns (uint);

    function createVault(address) external returns (address);
    function getVault(address) external view returns (address);
    function allVaults(uint) external view returns (address);
    function isVaultManager(address) external view returns (bool);
    function vaultsLength() external view returns (uint);

    function setVaultManager(address _manager, bool _status) external ;
    function setMaxProtocolFee(uint) external;
    function setProtocolFeeTo(address) external;
    function setMinRedemptionFee(uint) external;
    function setMaxRedemptionAdjustment(uint) external;
    function setRedemptionFeeTo(address) external;
}

// File: contracts/tokens/MintableBurnableERC20.sol



// SPDX-License-Identifier: GPL-3.0

pragma solidity 0.8.17;

import "@openzeppelin/contracts/token/ERC20/ERC20.sol";
import "../interfaces/IMintableBurnableERC20.sol";

contract MintableBurnableERC20 is ERC20, IMintableBurnableERC20 {

    address public owner;
    mapping (address => bool) public isMinter;

    constructor(string memory _name, string memory _symbol) ERC20(_name, _symbol) {
        owner = msg.sender;
    }

    function transferOwner(address _owner) external override {
        require(msg.sender == owner, "!owner");
        owner = _owner;
    }

    function setMinter(address _minter, bool _status) external override {
        require(msg.sender == owner, "!owner");
        isMinter[_minter] = _status;
    }

    function mint(address _to, uint256 _amount) external override {
        require(isMinter[msg.sender] == true, "!minter");
        _mint(_to, _amount);
    }

    function burn(uint256 _amount) external override {
        _burn(msg.sender, _amount);
    }
}

// File: contracts/Vault.sol



// SPDX-License-Identifier: GPL-3.0

pragma solidity 0.8.17;

import "./interfaces/IVault.sol";

contract Vault is IVault {

    /* ========== STATES ========== */

    IVaultFactory public immutable override factory;
    address public override owner;
    uint public override deposited;
    uint public override minted;

    /* ========== CONSTRUCTOR ========== */

    constructor() {
        factory = IVaultFactory(msg.sender);
    }

    // Called once by the factory on deployment
    function initialize(address _owner) external override {
        require(msg.sender == address(factory), '!factory');
        owner = _owner;
    }

    /* ========== VIEWS ========== */

    function availableBalance() public view override returns (uint) {
        return deposited - minted;
    }

    function pendingYield() public view override returns (uint) {
        uint balance = factory.collateral().balanceOf(address(this));
        return balance < deposited ? 0 : balance - deposited;
    }

    function mintRatio() public view override returns (uint) {
        if (deposited == 0) {
            return 0;
        }
        return minted * factory.PRECISION() / deposited;
    }

    // Calculate protocol fee based on mint ratio
    function protocolFee() public view override returns (uint) {
        return mintRatio() * factory.maxProtocolFee() / factory.PRECISION();
    }

    // Calculate redemption fee based on mint ratio
    function redemptionFee() public view override returns (uint) {
        return factory.minRedemptionFee() + factory.maxRedemptionAdjustment() * (factory.PRECISION() - mintRatio()) / factory.PRECISION();
    }

    /* ========== USER FUNCTIONS ========== */

    // Deposit collateral from msg.sender to vault
    function deposit(uint _amount) external override onlyManagerOrOwner() returns (uint) {
        claim();
        // stETH have known rounding error on transfers by 1-2 wei
        uint before = factory.collateral().balanceOf(address(this));
        factory.collateral().transferFrom(msg.sender, address(this), _amount);
        uint actualAmount = factory.collateral().balanceOf(address(this)) - before;
        deposited += actualAmount;
        emit Deposit(actualAmount);
        return actualAmount;
    }

    // Withdraw available collateral to owner
    function withdraw(uint _amount) external override onlyManagerOrOwner() {
        claim();
        _withdraw(_amount, owner);
        emit Withdraw(_amount);
    }

    // Mint token to vault owner using available collateral
    function mint(uint _amount) external override onlyManagerOrOwner() {
        claim();
        require(availableBalance() >= _amount, "!available");
        factory.token().mint(owner, _amount);
        minted += _amount;
        emit Mint(_amount);
    }

    // Burn token from msg.sender for vault
    function burn(uint _amount) external override onlyManagerOrOwner() {
        claim();
        _burn(_amount);
        emit Burn(_amount);
    }

    // Claim pending yield into deposited, pay protocol fee
    function claim() public override {
        uint yield = pendingYield();
        uint fee = yield * protocolFee() / factory.PRECISION();
        factory.collateral().transfer(factory.protocolFeeTo(), fee);
        deposited += yield - fee;
        emit Claim(yield - fee, fee);
    }

    // Redeem collateral from vault by burning token from msg.sender and paying redemption fee
    function redeem(uint _amount) external {
        uint fee = _amount * redemptionFee() / factory.PRECISION();
        claim();
        _burn(_amount);
        _withdraw(fee, factory.redemptionFeeTo());
        _withdraw(_amount - fee, msg.sender);
        emit Redeem(msg.sender, _amount, fee);
    }

    /* ========== INTERNAL FUNCTIONS ========== */

    function _burn(uint _amount) internal {
        require(minted >= _amount, "!minted");
        factory.token().transferFrom(msg.sender, address(this), _amount);
        factory.token().burn(_amount);
        minted -= _amount;
    }

    function _withdraw(uint _amount, address _recipient) internal {
        require(availableBalance() >= _amount, "!available");
        deposited -= _amount;
        factory.collateral().transfer(_recipient, _amount);
    }

    /* ========== MODIFIERS ========== */

    modifier onlyManagerOrOwner() {
        require(msg.sender == owner || factory.isVaultManager(msg.sender), "!allowed");
        _;
    }

    /* ========== EVENTS ========== */

    event Deposit(uint amount);
    event Withdraw(uint amount);
    event Mint(uint amount);
    event Burn(uint amount);
    event Claim(uint yieldAfterProtocolFee, uint protocolFee);
    event Redeem(address indexed redeemer, uint amount, uint redemptionFee);
}

// File: contracts/VaultFactory.sol



// SPDX-License-Identifier: GPL-3.0

pragma solidity 0.8.17;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/proxy/Clones.sol";
import "./interfaces/IVaultFactory.sol";
import "./tokens/MintableBurnableERC20.sol";
import "./Vault.sol";

contract VaultFactory is IVaultFactory, Ownable {

    IERC20 public immutable override collateral;
    IMintableBurnableERC20 public immutable override token;
    address public immutable vaultImplementation;

    address public override protocolFeeTo;
    uint public override maxProtocolFee;
    address public override redemptionFeeTo;
    uint public override minRedemptionFee;
    uint public override maxRedemptionAdjustment;
    uint public constant override PRECISION = 1e18;

    mapping(address => address) public override getVault;
    address[] public override allVaults;
    mapping(address => bool) public override isVaultManager;

    constructor(IERC20 _collateral, string memory _name, string memory _symbol) {
        collateral = _collateral;
        token = new MintableBurnableERC20(_name, _symbol);
        vaultImplementation = address(new Vault());
        protocolFeeTo = msg.sender;
        redemptionFeeTo = msg.sender;
        maxProtocolFee = 40e16; // default to 40%
        minRedemptionFee = 5e15; // default to 0.5%
        maxRedemptionAdjustment = 15e15; // default to 1.5%
    }

    function vaultsLength() external override view returns (uint) {
        return allVaults.length;
    }

    function createVault(address _owner) external override returns (address) {
        require(getVault[_owner] == address(0), 'exists');
        bytes32 salt = keccak256(abi.encodePacked(_owner));
        address vault = Clones.cloneDeterministic(vaultImplementation, salt);
        IVault(vault).initialize(_owner);
        token.setMinter(vault, true);
        getVault[_owner] = vault;
        allVaults.push(vault);
        emit VaultCreated(_owner, vault, allVaults.length - 1);
        return vault;
    }

    function setVaultManager(address _manager, bool _status) external override onlyOwner {
        isVaultManager[_manager] = _status;
        emit SetVaultManager(_manager, _status);
    }

    function setMinRedemptionFee(uint _minRedemptionFee) external override onlyOwner {
        require(_minRedemptionFee > 0 && _minRedemptionFee <= 10e16, "fee needs to be > 0 and < 10%");
        minRedemptionFee = _minRedemptionFee;
        emit SetMinRedemptionFee(_minRedemptionFee);
    }

    function setMaxRedemptionAdjustment(uint _maxRedemptionAdjustment) external override onlyOwner {
        require(_maxRedemptionAdjustment > 0 && _maxRedemptionAdjustment <= 10e16, "adjustment needs to be > 0 and < 10%");
        maxRedemptionAdjustment = _maxRedemptionAdjustment;
        emit SetMaxRedemptionAdjustment(_maxRedemptionAdjustment);
    }

    function setRedemptionFeeTo(address _redemptionFeeTo) external override onlyOwner {
        redemptionFeeTo = _redemptionFeeTo;
        emit SetRedemptionFeeTo(_redemptionFeeTo);
    }

    function setMaxProtocolFee(uint _maxProtocolFee) external override onlyOwner {
        require(_maxProtocolFee > 0 && _maxProtocolFee <= 1e18, "fee needs to be > 0 and < 100%");
        maxProtocolFee = _maxProtocolFee;
        emit SetMaxProtocolFee(_maxProtocolFee);
    }

    function setProtocolFeeTo(address _protocolFeeTo) external override onlyOwner {
        protocolFeeTo = _protocolFeeTo;
        emit SetProtocolFeeTo(_protocolFeeTo);
    }

    // Factory should not have any balances, allow rescuing of accidental transfers
    function rescue(address _token, address payable _recipient) external onlyOwner {
        if (_token == address(0)) {
            (bool sent, ) = _recipient.call{value: address(this).balance}("");
            require(sent, "Failed to send Ether");
        } else {
            uint _balance = IERC20(_token).balanceOf(address(this));
            IERC20(_token).transfer(_recipient, _balance);
        }
    }

    event VaultCreated(address indexed owner, address indexed vault, uint id);
    event SetVaultManager(address indexed manager, bool status);
    event SetMinRedemptionFee(uint minRedemptionFee);
    event SetMaxRedemptionAdjustment(uint maxRedemptionAdjustment);
    event SetRedemptionFeeTo(address indexed redemptionFeeTo);
    event SetMaxProtocolFee(uint maxProtocolFee);
    event SetProtocolFeeTo(address indexed protocolFeeTo);
}